
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 9.4.3.3a which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_9_4_3_3a
Test Group:
/11/UTRAN34_NISPC.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D14wk49 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Intel XMM 7260
Verification Status:
PASS
4 Corrections required for test case 9.4.3.3a
4.1 Introduction

This section describes the changes required to make test case 9.4.3.3a run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D14wk49 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 Function fl_TC_9_4_3_3a_Body 
	Record Name
	fl_TC_9_4_3_3a_Body

	Reason for change
	1) In the current TTCN implementation at Step 1, the UE is made to perform a LAU REJ. However, the Cell A needs to be configured correctly for this. This is not implemented correctly. Similarly, later in the testcase body the Cell B should be reconfigured correctly at Steps #33, #73, #137. Same applies to Cell A at Steps #84,#169. It needs to be reconfigured correctly.  
2) The testcase is quite a long one, and it will help if intermediate verdicts are added in the TTCN. This will help improve the traceablity of the execution.

3) In the current TTCN implementation at Step 6 the cause in LAU REJ is set to #17 (Network failure). After this, the expectation is that the UE will set the lcation update attempt counter to 4, and start timer T3212 and thus NOT perform a LAU procedure for 6 minutes. However, using the REJ cause #17 will not result in this , since there is no reference in 24.008 that suggests that the UE will set the attempt counter to 4, and start T3212 upon reception of this cause in a LAU REJ message. However, as per 24.008 Sec 4.4.49, “g)Location updating reject, other causes than those treated in subclause 4.4.4.7, and cases of MM cause #22, if considered as abnormal cases according to subclause 4.4.4.7 : Upon reception of the cause codes #22, # 95, # 96, # 97, # 99 and # 111 the MS should set the attempt counter to 4. The MS waits for release of the RR connection as specified in subclause 4.4.4.8, and then proceeds as specified below”. Also in the same section following can be seen :” If the attempt counter is smaller than 4, the mobile station shall memorize that timer T3211 (or, if the conditions for cause code #22 specified in subclause 4.4.4.7 or subclause 4.5.1.1 are met, shall start timer T3246 and not start timer T3211) is to be started when the RR connection is released, otherwise it shall memorize that timer T3212 (or, if the conditions for cause code #22 specified in subclause 4.4.4.7 or subclause 4.5.1.1 are met, shall start timer T3246 and not start timer T3212) is to be started when the RR connection is released.” So, it is required that instead of cause #17, cause #96 should be used at Step 6. This needs to be corrected. A Prose CR will be raised for this.

	Summary of change
	1) Called function f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellA, utran34_CellB); at Steps #1, #84, #169. and called function f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA); at steps #33, #73, #137.
2) Added intermediate pass verdicts as shown in screenshot below.
3) At step 6, Instead of using cause #17, used cause #96 for LAU REJ.

	Source of change
	UTRAN34_LocationUpdating.ttcn

	
	


Before:

	function fl_TC_9_4_3_3a_Body() runs on UTRAN_PTC
  {
    var UTRAN_CellInfo_Type v_CellInfoA, v_CellInfoB;
    var NAS_PlmnId v_PLMN_IdentityA, v_PLMN_IdentityB;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var START_Value v_START_Value_cs := '00000000000000000000'B;
    var RRC_DATA_IND v_RRC_DataInd, v_RRC_DataInd_ps;
    var float v_T3211_min := f_UG_SetTimerToleranceMin(15.0); // 24.008 cl. 11
    var float v_T3212_min;
    var float v_T30sec := 30.0; // absolute value of 30 sec
    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step5;  // used in steps 5, 89
    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step34; // used in steps 34, 140
    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step13; // used in all other steps using Location Update Request
    var NAS_Lac v_LAC_A, v_LAC_B;
    var template (present) LocAreaId v_LAI_A, v_LAI_B, v_LAI_Deleted;
    var octetstring v_TMSI := px_TMSI_Def;
    var TI v_TI_R;
    var TI v_TI_S;
    var B1_Type v_FollowOnRequest;
    var B7_Type v_CauUnassNumber := '0000001'B;
    var boolean v_LAU_received := false;
    var boolean v_RAU_received := false;
    var boolean v_Attach_received := false;
    timer t_WaitForGMM := 5.0;
    timer t_WaitForCMService := 10.0;
    v_CellInfoA := f_UTRAN_CellInfo_Get(utran34_CellA);
    // Variables to be used with cell A
    v_PLMN_IdentityA := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA));
    v_LAC_A := v_CellInfoA.lac;
    v_LAI_A := cr_LAI(omit, v_PLMN_IdentityA, v_LAC_A);
    // Variables to be used with cell B
    v_CellInfoB := f_UTRAN_CellInfo_Get(utran34_CellB);
    v_PLMN_IdentityB := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellB));
    v_LAC_B := v_CellInfoB.lac;
    v_LAI_B := cr_LAI(omit, v_PLMN_IdentityB, v_LAC_B);
    // Variables independent of Cell A or B
    v_T3212_min := (int2float(tsc_T3212)/1000.0) - 15.0; // 6 min - 15 seconds, see step 9, same for both cells
    v_LAI_Deleted := cr_LAI(omit, v_PLMN_IdentityA, tsc_LAC_Deleted); // same for both cells as they use the same PLMN
    // TMSI used throughout the test case is px_TMSI_Def; this is the same principle as applied in 9.4.3.3
    // new TMSI allocated will therefore always be the same; UE ack's this via TMSI Reallocation Complete
    // changing the TMSI value is tested in a specific test case
    v_U_LocationUpdatingRequest_Step5 :=  cr_U_LocationUpdatingRequest_9433a(?, // CKSN = initial value
                                                                             v_LAI_B, // LAI = b
                                                                             cr_MobileIdTMSI_lv(v_TMSI)); // mobile identity = TMSI
    v_U_LocationUpdatingRequest_Step34 := cr_U_LocationUpdatingRequest_9433a(?, // CKSN = initial value
                                                                             v_LAI_A, // LAI = a
                                                                             cr_MobileIdTMSI_lv(v_TMSI)); // mobile identity = TMSI
    v_U_LocationUpdatingRequest_Step13 := cr_U_LocationUpdatingRequest_9433a(tsc_KeySeqDeleted, // CKSN = no key available
                                                                             v_LAI_Deleted, // LAI = deleted LAI(the MCC and MNC hold the previous values, the LAC is coded FFFE)
                                                                             f_Imsi2MobileIdentity(px_IMSI_Def)); // mobile identity = IMSI
    //@siclog "Step 1" siclog@
    //Set the cell type of cell A to the "Serving cell". Set the cell type of cell B to the "non-suitable cell".
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationCellOFF, tsc_AttenuationCellOFF);
    //@siclog "Step 2-4" siclog@
    //RRC CONNECTION Establishment with cause: Registration. If PS mode: a ROUTING AREA UPDATE REQUEST should be rejected with the cause "GPRS services not allowed".
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 5-6" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = initial value, LAI = b, mobile station classmark 1 as given by the ICS and mobile identity = TMSI.
    //LOCATION UPDATING REJECT. IE Reject cause is set to #22 in table 10.5.95 of TS 24.008, causes #2, #3, #6, #11, #12, #13 and #15 being excluded.
    if (pc_PS == true) {
      // wait for RAU and LAU in any order and reject
      t_WaitForGMM.start;
      interleave { // interleave does not allow calling user defined functions (TTCN standard V4.5.1 cl. 20.4)
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                  tsc_RB3,
                                                  cr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, ?),
                                                                    cr_RAI(omit),
                                                                    cr_PTMSI_SignatureAny ifpresent,
                                                                    ?)))
          {
            v_RAU_received := true;
          }
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step5))
          {
            v_LAU_received := true;
          }
      }
    }
    else { // only CS
      // wait for LAU and reject
      U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step5));
      f_UTRAN34_SendLocationUpdatingReject(tsc_RejCau_NtwFailure); // @sic R5-144637 sic@
    }
    // if pc_PS == true both if statements will be executed
    if (v_RAU_received == true) {
      t_WaitForGMM.stop;
      f_UTRAN34_SendRoutingAreaUpdateReject(tsc_RejCau_GPRS_SrvNotAllowed);
      v_RAU_received := false;
    }
    if (v_LAU_received == true) {
      f_UTRAN34_SendLocationUpdatingReject(tsc_RejCau_NtwFailure); // @sic R5-144637 sic@
      v_LAU_received := false;
    }
    //@siclog "Step 7-8" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Step 9" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3212 (tolerance -15s; 45s) at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3212_min);
    //@siclog "Step 10-12" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 13" siclog@
    //LOCATION UPDATING REQUEST. location updating type: "normal location update", CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), mobile station classmark 1 as given by the ICS and mobile identity = IMSI.
    //@siclog "Step 14" siclog@
    //LOCATION UPDATING REJECT. IE Reject cause = #17 "network failure".
    fl_TC_9_4_3_3a_LU_Rejection(v_U_LocationUpdatingRequest_Step13, tsc_RejCau_NtwFailure);
    //@siclog "Step 15-16" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Step 17" siclog@
    // The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 18-20" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 21" siclog@
    //LOCATION UPDATING REQUEST. location updating type = "normal location update", CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE) mobile station classmark 1 as given by the ICS and mobile identity = IMSI.
    //@siclog "Step 22-23" siclog@
    //AUTHENTICATION REQUEST and RESPONSE. CKSN = initial CKSN
    //@siclog "Step 24-25" siclog@
    //SECURITY MODE COMMAND and COMPLETE.
    //@siclog "Step 26" siclog@
    //LOCATION UPDATING ACCEPT. IE mobile Identity = new TMSI.
    //@siclog "Step 27" siclog@
    //TMSI REALLOCATION COMPLETE
    f_UTRAN34_LocationUpdate(utran34_CellA, true, TMSI, v_U_LocationUpdatingRequest_Step13);
    //@siclog "Step 28-29" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Step 30" siclog@
    //Set the cell type of cell B to the "Serving cell". Set the cell type of cell A to the "non-suitable cell".
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationServingCell, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationCellOFF, tsc_AttenuationCellOFF);
    //@siclog "Step 31-33" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 34" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = initial value, LAI = a, mobile station classmark 1 as given by the ICS and mobile identity = TMSI.
    //@siclog "Step 35" siclog@
    //LOCATION UPDATING REJECT. IE Reject cause is set to #38 in table 10.5.95 of TS 24.008, causes #2, #3, #6, #11, #12, #13 and #15 being excluded.
    fl_TC_9_4_3_3a_LU_Rejection(v_U_LocationUpdatingRequest_Step34, tsc_RejCau_CallNotIdentified);
    //@siclog "Step 36-37" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    //@siclog "Step 38" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 39-41" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 42" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), Mobile Identity = IMSI.
    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13)) -> value v_RRC_DataInd;
    v_START_Value_cs := v_RRC_DataInd.ttcn_start;
    //@siclog "Step 43" siclog@
    //The SS modifies the scrambling code of DL DPCH for generating lower layer failure.
    //@siclog "Step 44" siclog@
    //CELL UPDATE. CCCH
    //@siclog "Step 45" siclog@
    //RRC CONNECTION RELEASE. CCCH
    //@siclog "Step 46" siclog@
    //The SS re-modifies the scrambling code of DL DPCH to the original one.
    f_UTRAN34_LowerLayerFailureCheck(utran34_CellB);
    //@siclog "Step 47" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 48-50" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 51" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), Mobile Identity = IMSI
    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13));
    //@siclog "Step 52-53" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    //@siclog "Step 54" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 55-57" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 58" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE) mobile station classmark 1 as given by the ICS and mobile identity = IMSI.
    //@siclog "Step 59-60" siclog@
    //SS performs step 35 and 36, and step 37.
    //LOCATION UPDATING REJECT. IE Reject cause is set to #38 in table 10.5.95 of TS 24.008, causes #2, #3, #6, #11, #12, #13 and #15 being excluded.
    fl_TC_9_4_3_3a_LU_Rejection(v_U_LocationUpdatingRequest_Step13, tsc_RejCau_CallNotIdentified);
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    if (pc_EmergSpeech == true) {
      // Cause the UE to originate Emergency call. The request is assumed to be triggered by AT command D.
      f_UT_RequestCSCall(UT, px_EmergencyCallNumber);
      //@siclog "Step 61-63" siclog@
      //RRC CONNECTION Establishment with cause: Emergency.
      f_UTRAN34_RRC_ConnEst(utran34_CellB, emergencyCall);
      //@siclog "Step 64" siclog@
      //CM SERVICE REQUEST. CM service type = Emergency call establishment; CKSN = no key available; Mobile Identity = IMSI.
      U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,
                                          tsc_RB3,
                                          cr_U_ServReq_MO_Emergency(tsc_KeySeqDeleted, f_Imsi2MobileIdentity(px_IMSI_Def))))-> value v_RRC_DataInd;
      v_START_Value_cs := v_RRC_DataInd.ttcn_start;
      f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value_cs);
      //@siclog "Step 65" siclog@
      //CM SERVICE ACCEPT
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                               tsc_RB3,
                               cs_U_CM_ServAcc));
      //@siclog "Step 66" siclog@
      //EMERGENCY SETUP
      U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,
                                            tsc_RB3,
                                            cr_U_ESetup(cr_BcapSpeechMO))) -> value v_RRC_DataInd;
      v_TI_R := v_RRC_DataInd.msg.esetup.ti;
      v_TI_S := v_TI_R;
      v_TI_S.tiFlag := '1'B;
      //@siclog "Step 67" siclog@
      //RELEASE COMPLETE. Cause = unassigned number
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                               tsc_RB3,
                               cs_U_ReleaseComplete(v_TI_S, -, cs_Cau(v_CauUnassNumber))));
      //@siclog "Step 68-69" siclog@
      //The SS waits for the disconnection of the main signalling link.
      f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    }
    //@siclog "Step 70" siclog@
    // If possible (see ICS) USIM detachment is performed. Otherwise if possible (see ICS) switch off is performed. Otherwise the power is removed.
    f_UT_SwitchOffUE(UT, true, CNF_REQUIRED);
    //@siclog "Step 71" siclog@
    // The UE shall not initiate an RRC connection establishment on cell A or on cell B. This is checked during 30 s.
    f_UTRAN34_RRC_RandAccFail(v_T30sec); // no IMSI Detach as there has been a LU rejection before
    //@siclog "Step 72" siclog@
    // Depending on what has been performed in step 70 the UE is brought back to operation. The subsequent GMM attach should be rejected if received in the PS mode.
    f_UT_SwitchOnUE(UT, true, CNF_REQUIRED);
    //@siclog "Step 73-75" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 76" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), Mobile Identity = IMSI.
    //@siclog "Step 77-78" siclog@
    //AUTHENTICATION REQUEST and RESPONSE. CKSN = initial CKSN
    //@siclog "Step 79-80" siclog@
    //SECURITY MODE COMMAND and COMPLETE.
    //@siclog "Step 81" siclog@
    //LOCATION UPDATING ACCEPT. IE mobile Identity = new TMSI.
    //@siclog "Step 82" siclog@
    //TMSI REALLOCATION COMPLETE
    if (pc_PS == true) {
      // wait for GMM Attach and LAU in any order and reject GMM Attach
      t_WaitForGMM.start;
      interleave { // interleave does not allow calling user defined functions (TTCN standard V4.5.1 cl. 20.4)
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt, tsc_I_AttachTypeCombined)),
                                                                                               cr_MobileIdAny(omit),
                                                                                               cr_RAI(omit),
                                                                                               ?))) -> value v_RRC_DataInd_ps
          {
            v_Attach_received := true;
          }
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13))  -> value v_RRC_DataInd
          {
            v_LAU_received := true;
          }
      }
    }
    else { // only CS
      // wait for LAU
      f_UTRAN34_LocationUpdate(utran34_CellB, true, TMSI, v_U_LocationUpdatingRequest_Step13);
    }
    // if pc_PS == true both if statements will be executed
    if (v_Attach_received == true) {
      t_WaitForGMM.stop;
      v_START_Value_ps := v_RRC_DataInd_ps.ttcn_start;
      f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
      U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                               tsc_RB3,
                               cs_U_AttachReject(tsc_RejCau_GPRS_SrvNotAllowed)));
    }
    if (v_LAU_received == true) {
      v_START_Value_ps := v_RRC_DataInd.ttcn_start;
      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran34_CellB, v_START_Value_ps, true, TMSI);
    }
    //@siclog "Step 83-84" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    //@siclog "Step 85" siclog@
    //Set the cell type of cell A to the "Serving cell". Set the cell type of cell B to the "non-suitable cell".
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationCellOFF, tsc_AttenuationCellOFF);
    //@siclog "Step 86-88" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 89" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = initial value, LAI = b, mobile station classmark 1 as given by the ICS and mobile identity = TMSI.
    //@siclog "Step 90" siclog@
    //LOCATION UPDATING REJECT. IE Reject cause is set to #38 in table 10.5.95 of TS 24.008, causes #2, #3, #6, #11, #12, #13 and #15 being excluded
    fl_TC_9_4_3_3a_LU_Rejection(v_U_LocationUpdatingRequest_Step5, tsc_RejCau_CallNotIdentified);
    //@siclog "Step 91-92" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Step 93" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 94-96" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 97" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), Mobile Identity = IMSI.
    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13)) -> value v_RRC_DataInd;
    v_START_Value_cs := v_RRC_DataInd.ttcn_start;
    //@siclog "Step 98" siclog@
    //The SS modifies the scrambling code of DL DPCH for generating lower layer failure.
    //@siclog "Step 99" siclog@
    //CELL UPDATE. CCCH
    //@siclog "Step 100" siclog@
    //RRC CONNECTION RELEASE. CCCH
    //@siclog "Step 101" siclog@
    //The SS re-modifies the scrambling code of DL DPCH to the original one.
    f_UTRAN34_LowerLayerFailureCheck(utran34_CellA);
    //@siclog "Step 102" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 103-105" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 106" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), Mobile Identity = IMSI.
    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13));
    //@siclog "Step 107-108" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Step 109" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 110-112" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 113" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE) mobile station classmark 1 as given by the ICS and mobile identity = IMSI.    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_LocationUpdatingRequest)) -> value v_RRC_DataInd
    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13)) -> value v_RRC_DataInd;
    v_START_Value_cs := v_RRC_DataInd.ttcn_start;
    //@siclog "Step 114" siclog@
    //The SS modifies the scrambling code of DL DPCH for generating lower layer failure.
    //@siclog "Step 115" siclog@
    //CELL UPDATE. CCCH
    //@siclog "Step 116" siclog@
    //RRC CONNECTION RELEASE. CCCH
    //@siclog "Step 116a" siclog@
    //The SS re-modifies the scrambling code of DL DPCH to the original one.
    f_UTRAN34_LowerLayerFailureCheck(utran34_CellA);
    //@siclog "Step 116b" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 117-119" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 120" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE) mobile station classmark 1 as given by the ICS and mobile identity = IMSI.
    //@siclog "Step 121-122" siclog@
    //AUTHENTICATION REQUEST and RESPONSE. CKSN = initial CKSN
    //@siclog "Step 123-124" siclog@
    //SECURITY MODE COMMAND and COMPLETE.
    //@siclog "Step 125" siclog@
    //LOCATION UPDATING ACCEPT. IE mobile Identity = new TMSI. If the location updating type in the LOCATION UPDATING REQUEST contains 'FOR', then IE Follow-on Proceed is included in the ACCEPT and steps 127 to 131 will be omitted.
    //@siclog "Step 126" siclog@
    //TMSI REALLOCATION COMPLETE
    v_FollowOnRequest := f_UTRAN34_LocationUpdate(utran34_CellA, true, TMSI, v_U_LocationUpdatingRequest_Step13);
    // If the location updating type in the LOCATION UPDATING REQUEST contains 'FOR', then IE Follow-on Proceed is included in the ACCEPT and steps 127 to 131 will be omitted.
    if (v_FollowOnRequest == '0'B) { // LOCATION UPDATING REQUEST does not contain 'FOR'; FOP included in ACCEPT by default !!
      //@siclog "Step 127-128" siclog@
      //The SS waits for the disconnection of the main signalling link.
      f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
      // UE is now "idle, updated" in cell A.
      // The UE may or may not have memorised the request for CM connection. The steps 129 to 135 are therefore optional for the MS.
      // The SS waits for 10 seconds to check if the UE attempts to make a CM service request, else it will directly go to step 136.
      t_WaitForCMService.start;
      alt {
        [] U_TM.check(receive(car_RRC_ConnReq(utran34_CellA,
                                              tsc_RB0,
                                              cr_108_RRC_ConnReq_r8(?))))
          {
            // there is an RRCConnectionRequest sent by the UE which can be received by TTCN-3 in a sub-sequent function
            //@siclog "Step 129-131" siclog@
            //RRC CONNECTION Establishment
            f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
            //@siclog "Step 132" siclog@
            //CM SERVICE REQUEST. CKSN = initial value, Mobile identity = TMSI.
            //@siclog "Step 133" siclog@
            //CM SERVICE REJECT. Cause #17 (network failure).
            //@siclog "Step 134-135" siclog@
            //The SS waits for the disconnection of the main signalling link.
            fl_TC_9_4_3_3a_Steps132_135();
          }
        [] t_WaitForCMService.timeout
          {
            // nothing to be done, directly go to step 136
          }
      }
    }
    else { // LOCATION UPDATING REQUEST contains 'FOR'
      //@siclog "Step 132" siclog@
      //CM SERVICE REQUEST. CKSN = initial value, Mobile identity = TMSI.
      //@siclog "Step 133" siclog@
      //CM SERVICE REJECT. Cause #17 (network failure).
      //@siclog "Step 134-135" siclog@
      //The SS waits for the disconnection of the main signalling link.
      fl_TC_9_4_3_3a_Steps132_135();
    }


After:

	function fl_TC_9_4_3_3a_Body() runs on UTRAN_PTC
  {
    var UTRAN_CellInfo_Type v_CellInfoA, v_CellInfoB;
    var NAS_PlmnId v_PLMN_IdentityA, v_PLMN_IdentityB;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var START_Value v_START_Value_cs := '00000000000000000000'B;
    var RRC_DATA_IND v_RRC_DataInd, v_RRC_DataInd_ps;
    var float v_T3211_min := f_UG_SetTimerToleranceMin(15.0); // 24.008 cl. 11
    var float v_T3212_min;
    var float v_T30sec := 30.0; // absolute value of 30 sec
    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step5;  // used in steps 5, 89
    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step34; // used in steps 34, 140
    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step13; // used in all other steps using Location Update Request
    var NAS_Lac v_LAC_A, v_LAC_B;
    var template (present) LocAreaId v_LAI_A, v_LAI_B, v_LAI_Deleted;
    var octetstring v_TMSI := px_TMSI_Def;
    var TI v_TI_R;
    var TI v_TI_S;
    var B1_Type v_FollowOnRequest;
    var B7_Type v_CauUnassNumber := '0000001'B;
    var boolean v_LAU_received := false;
    var boolean v_RAU_received := false;
    var boolean v_Attach_received := false;
    timer t_WaitForGMM := 5.0;
    timer t_WaitForCMService := 10.0;
    v_CellInfoA := f_UTRAN_CellInfo_Get(utran34_CellA);
    // Variables to be used with cell A
    v_PLMN_IdentityA := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA));
    v_LAC_A := v_CellInfoA.lac;
    v_LAI_A := cr_LAI(omit, v_PLMN_IdentityA, v_LAC_A);
    // Variables to be used with cell B
    v_CellInfoB := f_UTRAN_CellInfo_Get(utran34_CellB);
    v_PLMN_IdentityB := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellB));
    v_LAC_B := v_CellInfoB.lac;
    v_LAI_B := cr_LAI(omit, v_PLMN_IdentityB, v_LAC_B);
    // Variables independent of Cell A or B
    v_T3212_min := (int2float(tsc_T3212)/1000.0) - 15.0; // 6 min - 15 seconds, see step 9, same for both cells
    v_LAI_Deleted := cr_LAI(omit, v_PLMN_IdentityA, tsc_LAC_Deleted); // same for both cells as they use the same PLMN
    // TMSI used throughout the test case is px_TMSI_Def; this is the same principle as applied in 9.4.3.3
    // new TMSI allocated will therefore always be the same; UE ack's this via TMSI Reallocation Complete
    // changing the TMSI value is tested in a specific test case
    v_U_LocationUpdatingRequest_Step5 :=  cr_U_LocationUpdatingRequest_9433a(?, // CKSN = initial value
                                                                             v_LAI_B, // LAI = b
                                                                             cr_MobileIdTMSI_lv(v_TMSI)); // mobile identity = TMSI
    v_U_LocationUpdatingRequest_Step34 := cr_U_LocationUpdatingRequest_9433a(?, // CKSN = initial value
                                                                             v_LAI_A, // LAI = a
                                                                             cr_MobileIdTMSI_lv(v_TMSI)); // mobile identity = TMSI
    v_U_LocationUpdatingRequest_Step13 := cr_U_LocationUpdatingRequest_9433a(tsc_KeySeqDeleted, // CKSN = no key available
                                                                             v_LAI_Deleted, // LAI = deleted LAI(the MCC and MNC hold the previous values, the LAC is coded FFFE)
                                                                             f_Imsi2MobileIdentity(px_IMSI_Def)); // mobile identity = IMSI
    //@siclog "Step 1" siclog@
    //Set the cell type of cell A to the "Serving cell". Set the cell type of cell B to the "non-suitable cell".
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationCellOFF, tsc_AttenuationCellOFF);
    //@siclog "Step 2-4" siclog@
    //RRC CONNECTION Establishment with cause: Registration. If PS mode: a ROUTING AREA UPDATE REQUEST should be rejected with the cause "GPRS services not allowed".
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellA, utran34_CellB);//WA#9_4_3_3a
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 5-6" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = initial value, LAI = b, mobile station classmark 1 as given by the ICS and mobile identity = TMSI.
    //LOCATION UPDATING REJECT. IE Reject cause is set to #22 in table 10.5.95 of TS 24.008, causes #2, #3, #6, #11, #12, #13 and #15 being excluded.
    if (pc_PS == true) {
      // wait for RAU and LAU in any order and reject
      t_WaitForGMM.start;
      interleave { // interleave does not allow calling user defined functions (TTCN standard V4.5.1 cl. 20.4)
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                  tsc_RB3,
                                                  cr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, ?),
                                                                    cr_RAI(omit),
                                                                    cr_PTMSI_SignatureAny ifpresent,
                                                                    ?)))
          {
            v_RAU_received := true;
          }
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step5))
          {
            v_LAU_received := true;
          }
      }
    }
    else { // only CS
      // wait for LAU and reject
      U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step5));
      f_UTRAN34_SendLocationUpdatingReject(tsc_RejCau_NtwFailure); // @sic R5-144637 sic@
    }
    // if pc_PS == true both if statements will be executed
    if (v_RAU_received == true) {
      t_WaitForGMM.stop;
      f_UTRAN34_SendRoutingAreaUpdateReject(tsc_RejCau_GPRS_SrvNotAllowed);
      v_RAU_received := false;
    }
    if (v_LAU_received == true) {
      f_UTRAN34_SendLocationUpdatingReject(tsc_RejCau_InvalidMandatoryInfo/*WA#9_4_3_3atsc_RejCau_NtwFailure*/); // @sic R5-144637 sic@
      v_LAU_received := false;
    }
    //@siclog "Step 7-8" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Step 9" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3212 (tolerance -15s; 45s) at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3212_min);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 9");//WA#9_4_3_3a
    //@siclog "Step 10-12" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 13" siclog@
    //LOCATION UPDATING REQUEST. location updating type: "normal location update", CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), mobile station classmark 1 as given by the ICS and mobile identity = IMSI.
    //@siclog "Step 14" siclog@
    //LOCATION UPDATING REJECT. IE Reject cause = #17 "network failure".
    fl_TC_9_4_3_3a_LU_Rejection(v_U_LocationUpdatingRequest_Step13, tsc_RejCau_NtwFailure);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 13");//WA#9_4_3_3a
    //@siclog "Step 15-16" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 16");//WA#9_4_3_3a
    //@siclog "Step 17" siclog@
    // The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 18-20" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 21" siclog@
    //LOCATION UPDATING REQUEST. location updating type = "normal location update", CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE) mobile station classmark 1 as given by the ICS and mobile identity = IMSI.
    //@siclog "Step 22-23" siclog@
    //AUTHENTICATION REQUEST and RESPONSE. CKSN = initial CKSN
    //@siclog "Step 24-25" siclog@
    //SECURITY MODE COMMAND and COMPLETE.
    //@siclog "Step 26" siclog@
    //LOCATION UPDATING ACCEPT. IE mobile Identity = new TMSI.
    //@siclog "Step 27" siclog@
    //TMSI REALLOCATION COMPLETE
    f_UTRAN34_LocationUpdate(utran34_CellA, true, TMSI, v_U_LocationUpdatingRequest_Step13);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 27");//WA#9_4_3_3a
    //@siclog "Step 28-29" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Step 30" siclog@
    //Set the cell type of cell B to the "Serving cell". Set the cell type of cell A to the "non-suitable cell".
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationServingCell, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationCellOFF, tsc_AttenuationCellOFF);
    //@siclog "Step 31-33" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA);//WA#9_4_3_3a
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 34" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = initial value, LAI = a, mobile station classmark 1 as given by the ICS and mobile identity = TMSI.
    //@siclog "Step 35" siclog@
    //LOCATION UPDATING REJECT. IE Reject cause is set to #38 in table 10.5.95 of TS 24.008, causes #2, #3, #6, #11, #12, #13 and #15 being excluded.
    fl_TC_9_4_3_3a_LU_Rejection(v_U_LocationUpdatingRequest_Step34, tsc_RejCau_CallNotIdentified);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 35");//WA#9_4_3_3a
    //@siclog "Step 36-37" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    //@siclog "Step 38" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 39-41" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 41");//WA#9_4_3_3a
    //@siclog "Step 42" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), Mobile Identity = IMSI.
    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13)) -> value v_RRC_DataInd;
    v_START_Value_cs := v_RRC_DataInd.ttcn_start;
    //@siclog "Step 43" siclog@
    //The SS modifies the scrambling code of DL DPCH for generating lower layer failure.
    //@siclog "Step 44" siclog@
    //CELL UPDATE. CCCH
    //@siclog "Step 45" siclog@
    //RRC CONNECTION RELEASE. CCCH
    //@siclog "Step 46" siclog@
    //The SS re-modifies the scrambling code of DL DPCH to the original one.
    f_UTRAN34_LowerLayerFailureCheck(utran34_CellB);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 46");//WA#9_4_3_3a
    //@siclog "Step 47" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 48-50" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 51" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), Mobile Identity = IMSI
    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13));
    //@siclog "Step 52-53" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 53");//WA#9_4_3_3a
    //@siclog "Step 54" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 55-57" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 58" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE) mobile station classmark 1 as given by the ICS and mobile identity = IMSI.
    //@siclog "Step 59-60" siclog@
    //SS performs step 35 and 36, and step 37.
    //LOCATION UPDATING REJECT. IE Reject cause is set to #38 in table 10.5.95 of TS 24.008, causes #2, #3, #6, #11, #12, #13 and #15 being excluded.
    fl_TC_9_4_3_3a_LU_Rejection(v_U_LocationUpdatingRequest_Step13, tsc_RejCau_CallNotIdentified);
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 60");//WA#9_4_3_3a
    if (pc_EmergSpeech == true) {
      // Cause the UE to originate Emergency call. The request is assumed to be triggered by AT command D.
      f_UT_RequestCSCall(UT, px_EmergencyCallNumber);
      //@siclog "Step 61-63" siclog@
      //RRC CONNECTION Establishment with cause: Emergency.
      f_UTRAN34_RRC_ConnEst(utran34_CellB, emergencyCall);
      //@siclog "Step 64" siclog@
      //CM SERVICE REQUEST. CM service type = Emergency call establishment; CKSN = no key available; Mobile Identity = IMSI.
      U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,
                                          tsc_RB3,
                                          cr_U_ServReq_MO_Emergency(tsc_KeySeqDeleted, f_Imsi2MobileIdentity(px_IMSI_Def))))-> value v_RRC_DataInd;
      v_START_Value_cs := v_RRC_DataInd.ttcn_start;
      f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value_cs);
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 64");//WA#9_4_3_3a
      //@siclog "Step 65" siclog@
      //CM SERVICE ACCEPT
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                               tsc_RB3,
                               cs_U_CM_ServAcc));
      //@siclog "Step 66" siclog@
      //EMERGENCY SETUP
      U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,
                                            tsc_RB3,
                                            cr_U_ESetup(cr_BcapSpeechMO))) -> value v_RRC_DataInd;
      v_TI_R := v_RRC_DataInd.msg.esetup.ti;
      v_TI_S := v_TI_R;
      v_TI_S.tiFlag := '1'B;
      //@siclog "Step 67" siclog@
      //RELEASE COMPLETE. Cause = unassigned number
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                               tsc_RB3,
                               cs_U_ReleaseComplete(v_TI_S, -, cs_Cau(v_CauUnassNumber))));
      //@siclog "Step 68-69" siclog@
      //The SS waits for the disconnection of the main signalling link.
      f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    }
    //@siclog "Step 70" siclog@
    // If possible (see ICS) USIM detachment is performed. Otherwise if possible (see ICS) switch off is performed. Otherwise the power is removed.
    f_UT_SwitchOffUE(UT, true, CNF_REQUIRED);
    //@siclog "Step 71" siclog@
    // The UE shall not initiate an RRC connection establishment on cell A or on cell B. This is checked during 30 s.
    f_UTRAN34_RRC_RandAccFail(v_T30sec); // no IMSI Detach as there has been a LU rejection before
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 71");//WA#9_4_3_3a
    //@siclog "Step 72" siclog@
    // Depending on what has been performed in step 70 the UE is brought back to operation. The subsequent GMM attach should be rejected if received in the PS mode.
    f_UT_SwitchOnUE(UT, true, CNF_REQUIRED);
    //@siclog "Step 73-75" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA);//WA#9_4_3_3a
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 76" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), Mobile Identity = IMSI.
    //@siclog "Step 77-78" siclog@
    //AUTHENTICATION REQUEST and RESPONSE. CKSN = initial CKSN
    //@siclog "Step 79-80" siclog@
    //SECURITY MODE COMMAND and COMPLETE.
    //@siclog "Step 81" siclog@
    //LOCATION UPDATING ACCEPT. IE mobile Identity = new TMSI.
    //@siclog "Step 82" siclog@
    //TMSI REALLOCATION COMPLETE
    if (pc_PS == true) {
      // wait for GMM Attach and LAU in any order and reject GMM Attach
      t_WaitForGMM.start;
      interleave { // interleave does not allow calling user defined functions (TTCN standard V4.5.1 cl. 20.4)
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt, tsc_I_AttachTypeCombined)),
                                                                                               cr_MobileIdAny(omit),
                                                                                               cr_RAI(omit),
                                                                                               ?))) -> value v_RRC_DataInd_ps
          {
            v_Attach_received := true;
          }
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13))  -> value v_RRC_DataInd
          {
            v_LAU_received := true;
          }
      }
    }
    else { // only CS
      // wait for LAU
      f_UTRAN34_LocationUpdate(utran34_CellB, true, TMSI, v_U_LocationUpdatingRequest_Step13);
    }
    // if pc_PS == true both if statements will be executed
    if (v_Attach_received == true) {
      t_WaitForGMM.stop;
      v_START_Value_ps := v_RRC_DataInd_ps.ttcn_start;
      f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
      U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                               tsc_RB3,
                               cs_U_AttachReject(tsc_RejCau_GPRS_SrvNotAllowed)));
    }
    if (v_LAU_received == true) {
      v_START_Value_ps := v_RRC_DataInd.ttcn_start;
      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran34_CellB, v_START_Value_ps, true, TMSI);
    }
    //@siclog "Step 83-84" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 84");//WA#9_4_3_3a
    //@siclog "Step 85" siclog@
    //Set the cell type of cell A to the "Serving cell". Set the cell type of cell B to the "non-suitable cell".
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationCellOFF, tsc_AttenuationCellOFF);
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellA, utran34_CellB);//WA#9_4_3_3a
    //@siclog "Step 86-88" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 89" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = initial value, LAI = b, mobile station classmark 1 as given by the ICS and mobile identity = TMSI.
    //@siclog "Step 90" siclog@
    //LOCATION UPDATING REJECT. IE Reject cause is set to #38 in table 10.5.95 of TS 24.008, causes #2, #3, #6, #11, #12, #13 and #15 being excluded
    fl_TC_9_4_3_3a_LU_Rejection(v_U_LocationUpdatingRequest_Step5, tsc_RejCau_CallNotIdentified);
    //@siclog "Step 91-92" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 92");//WA#9_4_3_3a
    //@siclog "Step 93" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 94-96" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 97" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), Mobile Identity = IMSI.
    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13)) -> value v_RRC_DataInd;
    v_START_Value_cs := v_RRC_DataInd.ttcn_start;
    //@siclog "Step 98" siclog@
    //The SS modifies the scrambling code of DL DPCH for generating lower layer failure.
    //@siclog "Step 99" siclog@
    //CELL UPDATE. CCCH
    //@siclog "Step 100" siclog@
    //RRC CONNECTION RELEASE. CCCH
    //@siclog "Step 101" siclog@
    //The SS re-modifies the scrambling code of DL DPCH to the original one.
    f_UTRAN34_LowerLayerFailureCheck(utran34_CellA);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 101");//WA#9_4_3_3a
    //@siclog "Step 102" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 103-105" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 106" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE), Mobile Identity = IMSI.
    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13));
    //@siclog "Step 107-108" siclog@
    //The SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 108");//WA#9_4_3_3a
    //@siclog "Step 109" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    //@siclog "Step 110-112" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 113" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE) mobile station classmark 1 as given by the ICS and mobile identity = IMSI.    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_LocationUpdatingRequest)) -> value v_RRC_DataInd
    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13)) -> value v_RRC_DataInd;
    v_START_Value_cs := v_RRC_DataInd.ttcn_start;
    //@siclog "Step 114" siclog@
    //The SS modifies the scrambling code of DL DPCH for generating lower layer failure.
    //@siclog "Step 115" siclog@
    //CELL UPDATE. CCCH
    //@siclog "Step 116" siclog@
    //RRC CONNECTION RELEASE. CCCH
    //@siclog "Step 116a" siclog@
    //The SS re-modifies the scrambling code of DL DPCH to the original one.
    f_UTRAN34_LowerLayerFailureCheck(utran34_CellA);
    //@siclog "Step 116b" siclog@
    //The UE shall not initiate an RRC connection establishment on cell A or on cell B during T3211 seconds at least after the RRC connection is released.
    f_UTRAN34_RRC_RandAccFail(v_T3211_min);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 116b");//WA#9_4_3_3a
    //@siclog "Step 117-119" siclog@
    //RRC CONNECTION Establishment with cause: Registration.
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 120" siclog@
    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = no key available, LAI = deleted LAI (the MCC and MNC hold the previous values, the LAC is coded FFFE) mobile station classmark 1 as given by the ICS and mobile identity = IMSI.
    //@siclog "Step 121-122" siclog@
    //AUTHENTICATION REQUEST and RESPONSE. CKSN = initial CKSN
    //@siclog "Step 123-124" siclog@
    //SECURITY MODE COMMAND and COMPLETE.
    //@siclog "Step 125" siclog@
    //LOCATION UPDATING ACCEPT. IE mobile Identity = new TMSI. If the location updating type in the LOCATION UPDATING REQUEST contains 'FOR', then IE Follow-on Proceed is included in the ACCEPT and steps 127 to 131 will be omitted.
    //@siclog "Step 126" siclog@
    //TMSI REALLOCATION COMPLETE
    v_FollowOnRequest := f_UTRAN34_LocationUpdate(utran34_CellA, true, TMSI, v_U_LocationUpdatingRequest_Step13);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 126");//WA#9_4_3_3a
    // If the location updating type in the LOCATION UPDATING REQUEST contains 'FOR', then IE Follow-on Proceed is included in the ACCEPT and steps 127 to 131 will be omitted.
    if (v_FollowOnRequest == '0'B) { // LOCATION UPDATING REQUEST does not contain 'FOR'; FOP included in ACCEPT by default !!
      //@siclog "Step 127-128" siclog@
      //The SS waits for the disconnection of the main signalling link.
      f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
      // UE is now "idle, updated" in cell A.
      // The UE may or may not have memorised the request for CM connection. The steps 129 to 135 are therefore optional for the MS.
      // The SS waits for 10 seconds to check if the UE attempts to make a CM service request, else it will directly go to step 136.
      t_WaitForCMService.start;
      alt {
        [] U_TM.check(receive(car_RRC_ConnReq(utran34_CellA,
                                              tsc_RB0,
                                              cr_108_RRC_ConnReq_r8(?))))
          {
            // there is an RRCConnectionRequest sent by the UE which can be received by TTCN-3 in a sub-sequent function
            //@siclog "Step 129-131" siclog@
            //RRC CONNECTION Establishment
            f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
            //@siclog "Step 132" siclog@
            //CM SERVICE REQUEST. CKSN = initial value, Mobile identity = TMSI.
            //@siclog "Step 133" siclog@
            //CM SERVICE REJECT. Cause #17 (network failure).
            //@siclog "Step 134-135" siclog@
            //The SS waits for the disconnection of the main signalling link.
            fl_TC_9_4_3_3a_Steps132_135();
          }
        [] t_WaitForCMService.timeout
          {
            // nothing to be done, directly go to step 136
          }
      }
    }
    else { // LOCATION UPDATING REQUEST contains 'FOR'
      //@siclog "Step 132" siclog@
      //CM SERVICE REQUEST. CKSN = initial value, Mobile identity = TMSI.
      //@siclog "Step 133" siclog@
      //CM SERVICE REJECT. Cause #17 (network failure).
      //@siclog "Step 134-135" siclog@
      //The SS waits for the disconnection of the main signalling link.
      fl_TC_9_4_3_3a_Steps132_135();
    }


4.3 Function TC_9_4_3_3a()
	Record Name
	TC_9_4_3_3a()

	Reason for change
	In the current TTCN implementation the guard timer is set to 300s, which is very insufficient. This needs to be increased to 1200s. 

	Summary of change
	Increased the guard timer to 1200s. Please see screenshot. 

	Source of change
	UTRAN_Testsuite.ttcn

	
	


Before:

	testcase TC_9_4_3_3a() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Location updating / abnormal cases / attempt counter equal to 4
    var UTRAN_PTC    v_UTRAN      := null;
    var GERAN_PTC    v_GERAN      := null;
    var EUTRA_PTC    v_EUTRA      := null;
    timer t_GuardTimer := int2float(300);  
    v_UTRAN := UTRAN_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_9_4_3_3a_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }


After:

	testcase TC_9_4_3_3a() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Location updating / abnormal cases / attempt counter equal to 4
    var UTRAN_PTC    v_UTRAN      := null;
    var GERAN_PTC    v_GERAN      := null;
    var EUTRA_PTC    v_EUTRA      := null;
    timer t_GuardTimer := int2float(1200);//WA#9_4_3_3a : Changed guard timer 300-> 1200
    v_UTRAN := UTRAN_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_9_4_3_3a_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }


4.4 Function f_UTRAN34_LowerLayerFailureCheck
	Record Name
	cr_108_CellUpdate_r10_9_4_3_3a

	Reason for change
	As per the Prose, at Steps #44, #99, #115, #150, #174 the UE shuold send a CellUpdate. In the current TTCN, the function he purpose of function f_UTRAN34_LowerLayerFailureCheck is called to capture the CellUpdate. However, this function uses the template cr_108_CellUpdate_r10 which expects the va40NonCriticalExtensions. However, as per 25.331 Sec 11.2 PDU definitions,  The va40NonCriticalExtensions can be omitted if its inclusion would result in the size of the message exceeding the maximum PDU size possible for CCCH transmission. So, with the current implementation, a conformant UE will fail the testcase. This needs to be changed.

	Summary of change
	Created a new template cr_108_CellUpdate_r10_9_4_3_3a and used it in the function f_UTRAN34_LowerLayerFailureCheck. Please see screenshots.

	Source of change
	UTRAN34_LowerLayerFailureFunctions.ttcn

	
	


Before:

	function f_UTRAN34_LowerLayerFailureCheck(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC
  {
    var CellUpdate v_CellUpdate;
    var UTRAN_CellInfo_Type v_CellInfo := f_UTRAN_CellInfo_Get(p_CellId);
    timer t_RadioLinkFailure := tsc_RadioLinkFailure; // 13500 ms
    // Reconfigure SS with a scrambling code of downlink transmission (DL DPCH) different from the default one
    f_SS_LowerLayerFailure(p_CellId,
                           tsc_DL_DPCH1_2ndScrC + 1);
    // SS Receives Cell Update
    // Start Radio Link Failure timer
    t_RadioLinkFailure.start;
    v_CellUpdate := f_UTRAN_RRC_ReceiveCellUpdateNonPeriodic( p_CellId,
                                                                                                       cr_108_CellUpdate_r10(v_CellInfo.uRNTI, radiolinkFailure));
    t_RadioLinkFailure.stop;
    // SS sends RRC CONNECTION RELEASE message
    U_UM.send(cas_RRC_ConnRelCCCH(p_CellId,
                                  tsc_RB0,
                                  f_UTRAN34_Get_cs_108_RRC_ConnRelCCCH(p_CellId)));
    f_UTRAN34_CRLC_RelReconfSRB(p_CellId);
    f_UTRAN_CellInfo_SetConfigType(p_CellId, cell_DCH_StandAloneSRB_NoConn);
    // Reconfigure SS with the default scrambling code
    f_SS_LowerLayerFailure(p_CellId,
                           tsc_DL_DPCH1_2ndScrC);
  }


After:

	function f_UTRAN34_LowerLayerFailureCheck(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC
  {
    var CellUpdate v_CellUpdate;
    var UTRAN_CellInfo_Type v_CellInfo := f_UTRAN_CellInfo_Get(p_CellId);
    timer t_RadioLinkFailure := tsc_RadioLinkFailure; // 13500 ms
    // Reconfigure SS with a scrambling code of downlink transmission (DL DPCH) different from the default one
    f_SS_LowerLayerFailure(p_CellId,
                           tsc_DL_DPCH1_2ndScrC + 1);
    // SS Receives Cell Update
    // Start Radio Link Failure timer
    t_RadioLinkFailure.start;
    v_CellUpdate := f_UTRAN_RRC_ReceiveCellUpdateNonPeriodic( p_CellId,
                         /*WA#9_4_3_3a cr_108_CellUpdate_r10*/cr_108_CellUpdate_r10_9_4_3_3a(v_CellInfo.uRNTI, radiolinkFailure));
    t_RadioLinkFailure.stop;
    // SS sends RRC CONNECTION RELEASE message
    U_UM.send(cas_RRC_ConnRelCCCH(p_CellId,
                                  tsc_RB0,
                                  f_UTRAN34_Get_cs_108_RRC_ConnRelCCCH(p_CellId)));
    f_UTRAN34_CRLC_RelReconfSRB(p_CellId);
    f_UTRAN_CellInfo_SetConfigType(p_CellId, cell_DCH_StandAloneSRB_NoConn);
    // Reconfigure SS with the default scrambling code
    f_SS_LowerLayerFailure(p_CellId,
                           tsc_DL_DPCH1_2ndScrC);
  }


4.5 Template cr_108_CellUpdate_r10_9_4_3_3a  
	Record Name
	cr_108_CellUpdate_r10_9_4_3_3a  

	Reason for change
	Please Change above for function f_UTRAN34_LowerLayerFailureCheck.

	Summary of change
	Created a new template cr_108_CellUpdate_r10_9_4_3_3a.
Please see screenshot. 

	Source of change
	UTRAN34_RRC_Templates.ttcn

	
	


	template (present) UL_CCCH_Message cr_108_CellUpdate_r10_9_4_3_3a(U_RNTI p_U_RNTI,
                                                           CellUpdateCause p_CellUpdateCause) :=
  {
    integrityCheckInfo := *,
    message_ := {
      cellUpdate  := {
        u_RNTI                              := p_U_RNTI,
        startList                           :=?,
        am_RLC_ErrorIndicationRb2_3or4      := false,
        am_RLC_ErrorIndicationRb5orAbove    := false,
        cellUpdateCause                     := p_CellUpdateCause,
        failureCause                        := omit,
        rb_timer_indicator                  := {
          t314_expired    := false,
          t315_expired    := false
        },
        measuredResultsOnRACH               := *,
        laterNonCriticalExtensions          := {
          cellUpdate_r3_add_ext           := *,
          v590NonCriticalExtensions       := {
            cellUpdate_v590ext          := {
              establishmentCause := omit
            },
            v690NonCriticalExtensions   := {
              cellUpdate_v690ext          := {
                cellUpdateCause_ext     := *,
                trafficVolumeIndicator  := omit,
                measuredResultsOnRACHinterFreq := *,
                reconfigurationStatusIndicator := omit
              },
              v6b0NonCriticalExtensions       := {
                cellUpdate_v6b0ext              := {
                  mbmsSelectedServices :=  omit
                },
                v770NonCriticalExtensions       := {
                  cellUpdate_v770ext            := {
                    csCallType                  := omit,
                    hspdschReception_CellFach   := *,
                    ueMobilityStateIndicator    := omit,
                    capabilityChangeIndicator   := omit
                  },
                  v860NonCriticalExtensions     := {
                    cellUpdate_v860ext          := {
                      supportOfCommonEDCH                 := *,
                      supportOfHS_DSCHDRXOperation        := *,
                      supportOfMACiis                     := *,
                      supportOfSPSOperation               := *,
                      supportOfControlChannelDRXOperation := *
                    },
                    va40NonCriticalExtensions  := *     
                    /*As per 25.331 Sec 11.2 PDU definitions,  The va40NonCriticalExtensions can be omitted if its inclusion





 would result in the size of the message exceeding the maximum PDU size possible for CCCH transmission
                    := {
                      cellUpdate_va40ext          := {
                        securityRevertStatusIndicator   := *,
                        loggedMeasAvailable             := *,
                        loggedANRResultsAvailable       := *
                      },
                      vb50NonCriticalExtensions       := omit
                    }*/
                  }
                }
              }
            }
          }
        }
      }
    }
  };


4.6 Function f_UTRAN34_SendRoutingAreaUpdateReject
	Record Name
	f_UTRAN34_SendRoutingAreaUpdateReject

	Reason for change
	The function f_UTRAN34_SendRoutingAreaUpdateReject is used to send a RAU REJ to the UE during Steps 2 to 10. The current implementation of this function uses template cas_Datareq_CS. This is incorrect and needs to be changed.

	Summary of change
	Instead of using cas_Datareq_CS, used cas_Datareq_PS in the function f_UTRAN34_SendRoutingAreaUpdateReject. Please see the screenshot. 

	Source of change
	UTRAN34_Commonfunctions.ttcn

	
	


Before:

	function f_UTRAN34_SendRoutingAreaUpdateReject(RejCau p_RejCau) runs on UTRAN_PTC
  {
    U_Dc.send (cas_DataReq_CS(utran_CellDedicated,
                              tsc_RB3,
                              cs_U_RAUpdReject(cs_GMM_Cause(oct2bit(p_RejCau)))));
  }


After:

	function f_UTRAN34_SendRoutingAreaUpdateReject(RejCau p_RejCau) runs on UTRAN_PTC
  {
    U_Dc.send (cas_DataReq_PS/*WA#9_4_3_3a cas_DataReq_CS*/(utran_CellDedicated,
                              tsc_RB3,
                              cs_U_RAUpdReject(cs_GMM_Cause(oct2bit(p_RejCau)))));
  }


4.7 Function f_TC_9_4_3_3a_UTRAN()
	Record Name
	f_TC_9_4_3_3a_UTRAN()

	Reason for change
	In the current TTCN implementation of this function, the function f_UTRAN34_Postamble(utran34_CellA, U1_IDLE)  is called. This expects the UE to perfrom a PS DETACH. However, in this testcase the UE has never been attached for PS services and will only perform an IMSI DETACH. This needs to be corrected.

	Summary of change
	Created a new function Called f_UTRAN34_SwitchPower_Off_IMSI_DetachOnly  and used it instead of the function f_UTRAN34_Postamble(utran34_CellA, U1_IDLE).   

please see the screenshots below.

	Source of change
	UTRAN34_LocationUpdating.ttcn

	
	


Before:

	function f_TC_9_4_3_3a_UTRAN() runs on UTRAN_PTC
  {
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Initialization of Cell B per Initial conditions
    // Info for cell B
    f_UTRAN34_CellInfo_SetAttFlag(utran34_CellB, tsc_AttOn);
    f_UTRAN34_CellInfo_SetT3212(utran34_CellB, tsc_T3212_1);  // 6 minutes, see initial conditions
    f_UTRAN_CellInfo_SetLAC(utran34_CellB, tsc_LAC_2);
    f_UTRAN_CellInfo_SetNMO(utran34_CellB, tsc_NMO_II);
    // CellB started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_CellB);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellB);
    // Initialization of Cell A per Initial conditions
    // Info for cell A
    f_UTRAN34_CellInfo_SetAttFlag(utran34_CellA, tsc_AttOn);
    f_UTRAN34_CellInfo_SetT3212(utran34_CellA, tsc_T3212_1);  // 6 minutes, see initial conditions
    f_UTRAN_CellInfo_SetNMO(utran34_CellA, tsc_NMO_II);
    // CellA started switched off
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    // Set attenuation, switch on UE, and Idle update to bring UE to Cell_DCH state to register
    // The UE is "Idle updated" on cell B with a valid CKSN and a TMSI.
    f_UTRAN34_Preamble(utran34_CellB);
    f_UTRAN_TestBody_Set(true);
    fl_TC_9_4_3_3a_Body();
    f_UTRAN_TestBody_Set(false);
    // Release RRC connection on cell 1 and release the cells
    f_UTRAN34_Postamble(utran34_CellA, U1_IDLE);

    f_UTRAN_ReleaseCell(utran34_CellB, f_UTRAN_CellInfo_GetConfigType(utran34_CellB));
  } // f_TC_9_4_3_3a_UTRAN


After:

	function f_TC_9_4_3_3a_UTRAN() runs on UTRAN_PTC
  {
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Initialization of Cell B per Initial conditions
    // Info for cell B
    f_UTRAN34_CellInfo_SetAttFlag(utran34_CellB, tsc_AttOn);
    f_UTRAN34_CellInfo_SetT3212(utran34_CellB, tsc_T3212_1);  // 6 minutes, see initial conditions
    f_UTRAN_CellInfo_SetLAC(utran34_CellB, tsc_LAC_2);
    f_UTRAN_CellInfo_SetNMO(utran34_CellB, tsc_NMO_II);
    // CellB started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_CellB);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellB);
    // Initialization of Cell A per Initial conditions
    // Info for cell A
    f_UTRAN34_CellInfo_SetAttFlag(utran34_CellA, tsc_AttOn);
    f_UTRAN34_CellInfo_SetT3212(utran34_CellA, tsc_T3212_1);  // 6 minutes, see initial conditions
    f_UTRAN_CellInfo_SetNMO(utran34_CellA, tsc_NMO_II);
    // CellA started switched off
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    // Set attenuation, switch on UE, and Idle update to bring UE to Cell_DCH state to register
    // The UE is "Idle updated" on cell B with a valid CKSN and a TMSI.
    f_UTRAN34_Preamble(utran34_CellB);
    f_UTRAN_TestBody_Set(true);
    fl_TC_9_4_3_3a_Body();
    f_UTRAN_TestBody_Set(false);
    // Release RRC connection on cell 1 and release the cells
    f_UTRAN34_SwitchPower_Off_IMSI_DetachOnly(utran34_CellA, U1_IDLE, NotSpecified);//WA#9_4_3_3a 
    //WA#9_4_3_3a f_UTRAN34_Postamble(utran34_CellA, U1_IDLE);
    f_UTRAN_ReleaseCell(utran34_CellB, f_UTRAN_CellInfo_GetConfigType(utran34_CellB));
  } // f_TC_9_4_3_3a_UTRAN


4.8 Function f_UTRAN34_SwitchPower_Off_IMSI_DetachOnly 
	Record Name
	f_UTRAN34_SwitchPower_Off_IMSI_DetachOnly

	Reason for change
	Please see the change above for function f_TC_9_4_3_3a_UTRAN() .

	Summary of change
	Created a new function Called f_UTRAN34_SwitchPower_Off_IMSI_DetachOnly.  

please see the screenshots below.

	Source of change
	UTRAN34_LocationUpdating.ttcn

	
	


	function f_UTRAN34_SwitchPower_Off_IMSI_DetachOnly(UTRAN_CellId_Type         p_CellId,
                                     UTRA_POSTAMBLE_STATE_Type p_Utra_Postamble_State,
                                     UTRAN_GERAN_ATTACH_Type p_AttachType,
                                     boolean p_USIM_Rmvd := false) runs on UTRAN_PTC
  {
    var RB_ConfigType v_CellConfig := f_UTRAN_CellInfo_GetConfigType(p_CellId);
    var integer v_AttFlag := f_UTRAN_CellInfo_GetAttFlag(p_CellId);
    var RRC_Rel_Status v_RRC_Rel_Status := cell_Dch;
    var UTRA_POSTAMBLE_CONDITIONS v_Utra_Postamble_Condition := f_UTRA_SetPostambleCondition(p_AttachType); // @sic R5s110176 sic@
    var B2_Type v_Nmo := f_UTRAN_CellInfo_GetNMO(p_CellId);
    var RB_ConfigType v_CellConfig := f_UTRAN_CellInfo_GetConfigType(p_CellId);
    if (v_CellConfig == cell_FACH or v_CellConfig == cell_FACH_NoConn) {
      v_RRC_Rel_Status := cell_Fach_Dcch;
    }
    //Step 1: The UE can be switched off or USIM removed
    if (pc_SwitchOnOff or p_USIM_Rmvd and pc_USIM_Removal) { // @sic R5s120597r1 sic@
      if (p_Utra_Postamble_State == U2_CONNECTED) { // @sic R5s140611 sic@
        f_UTRAN34_RRC_ConnRel(p_CellId, cell_Dch);
      }
      //Step 1: The UE is switched off
      f_UT_SwitchOffUE(UT, p_USIM_Rmvd);
      //Step 2-4: RRC Connection establishment as UE is always in UTRA idle end state
      f_UTRAN34_RRC_ConnEst(p_CellId, detach);
      U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_Imsi_DetachInd)); 
      f_UTRAN34_RRC_ConnRel_AfterSwitchOff(p_CellId, v_RRC_Rel_Status);
    }    
    else {
      f_UT_PowerOffUE (UT);
    }
    f_UTRAN_ReleaseCell(p_CellId, v_CellConfig); // @sic R5s110176 sic@
  } // f_UTRAN34_SwitchPower_Off


4.9 Constant tsc_RejCau_InvalidMandatoryInfo
	Record Name
	tsc_RejCau_InvalidMandatoryInfo

	Reason for change
	As explained in point 3 of change 4.2, the REJ cause should be set to #96. The current TTN does not implement this.This needs to be implemented.

	Summary of change
	Created a new constant tsc_RejCau_InvalidMandatoryInfo  and assigned it value #96. Please see screenshot.

	Source of change
	UTRAN34_LocationUpdating.ttcn

	
	


	//WA#9_4_3_3a : added below causes for 9_4_3_3a
  const RejCau      tsc_RejCau_InvalidMandatoryInfo := '60'O;     //Reject cause: Invalid Mandatory information #96


5 Branches executed in test case 9.4.3.3a
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 Intel XMM 7260

The Intel XMM 7260 UE passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_9_4_3_3a_Intel_XMM_7260.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
In the log provided, masking is applied to prevent any disclosure of confidential information.
7 References
	[1]
	R5s141391
This archive comprises Execution log and PICS/PIXIT information.


